

Definition and Elementary Properties of Ultrametric Spaces

Benoît Ballenghien* Benjamin Puyobro† Burkhardt Wolff‡

February 7, 2026

Abstract

An ultrametric space is a metric space in which the triangle inequality is strengthened by using the maximum instead of the sum. More formally, such a space is equipped with a real-valued application *dist*, called distance, verifying the four following conditions.

$$\begin{aligned} \text{dist } x \ y &\geq 0 \\ \text{dist } x \ y &= \text{dist } y \ x \\ \text{dist } x \ y = 0 &\iff x = y \\ \text{dist } x \ z &\leq \max (\text{dist } x \ y) (\text{dist } y \ z) \end{aligned}$$

In this entry, we present an elementary formalization of these spaces relying on axiomatic type classes. The connection with standard metric spaces is obtained through a subclass relationship, and fundamental properties of ultrametric spaces are formally established.

*Université Paris-Saclay, CNRS, ENS Paris-Saclay, LMF
<https://orcid.org/0009-0000-4941-187X>

†Université Paris-Saclay, IRT SystemX, CNRS, ENS Paris-Saclay, LMF
<https://orcid.org/0009-0006-0294-9043>

This work has been supported by the French government under the “France 2030” program, as part of the SystemX Technological Research Institute within the CVH project.

‡Université Paris-Saclay, CNRS, ENS Paris-Saclay, LMF
<https://orcid.org/0000-0002-9648-7663>

Contents

1	Definition	2
2	Properties on Balls	2
2.1	Balls are centered everywhere	3
2.2	Balls are “clopen”	3
2.3	Balls are disjoint or contained	4
2.4	Distance to a Ball	6
3	Additional Properties	7
3.1	Cauchy Sequences	7
3.2	Isosceles Triangle Principle	9
3.3	Distance to a convergent Sequence	9
3.4	Diameter	9
3.5	Totally disconnected	10

1 Definition

setup $\langle \text{Sign.add-const-constraint } (\mathbf{const-name} \langle \text{dist} \rangle, \text{NONE}) \rangle$
— To be able to use *dist* out of the *metric-space* class.

```
class ultrametric-space = uniformity-dist + open-uniformity +  
  assumes dist-eq-0-iff [simp]:  $\langle \text{dist } x \ y = 0 \longleftrightarrow x = y \rangle$   
  and ultrametric-dist-triangle2:  $\langle \text{dist } x \ y \leq \max (\text{dist } x \ z) (\text{dist } y \ z) \rangle$ 
```

begin

```
subclass metric-space
```

```
proof (unfold-locales)
```

```
  show  $\langle \text{dist } x \ y = 0 \longleftrightarrow x = y \rangle$  for x y by simp
```

```
next
```

```
  show  $\langle \text{dist } x \ y \leq \text{dist } x \ z + \text{dist } y \ z \rangle$  for x y z
```

```
    by (rule order-trans[OF ultrametric-dist-triangle2, of - z], simp)  
      (metis local.dist-eq-0-iff ultrametric-dist-triangle2 max.idem)
```

```
qed
```

```
end
```

setup $\langle \text{Sign.add-const-constraint } (\mathbf{const-name} \langle \text{dist} \rangle, \text{SOME } \mathbf{typ} \langle 'a \rangle$
 $:: \text{metric-space} \Rightarrow 'a \Rightarrow \text{real} \rangle) \rangle$

— Back to normal.

```
class complete-ultrametric-space = ultrametric-space +
```

```
  assumes Cauchy-convergent:  $\langle \text{Cauchy } X \Longrightarrow \text{convergent } X \rangle$ 
```

```

begin

subclass complete-space by unfold-locales (fact Cauchy-convergent)

end

```

2 Properties on Balls

In ultrametric space, balls satisfy very strong properties.

```

context ultrametric-space begin

```

```

lemma ultrametric-dist-triangle: ⟨dist x z ≤ max (dist x y) (dist y z)⟩
  using ultrametric-dist-triangle2 [of x z y] by (simp add: dist-commute)

```

```

lemma ultrametric-dist-triangle3: ⟨dist x y ≤ max (dist a x) (dist a
y)⟩
  using ultrametric-dist-triangle2 [of x y a] by (simp add: dist-commute)

```

```

end

```

2.1 Balls are centered everywhere

```

context fixes x :: ⟨'a :: ultrametric-space⟩ begin

```

The best way to do this would be to work in the context *ultrametric-space*. Unfortunately, *ball*, *cball*, etc. are not defined inside the context *metric-space* but through a sort constraint.

```

lemma ultrametric-every-point-of-ball-is-centre :
  ⟨ball y r = ball x r⟩ if ⟨y ∈ ball x r⟩
proof (unfold set-eq-iff, rule allI)
  from ⟨y ∈ ball x r⟩ have * : ⟨dist x y < r⟩ by simp
  show ⟨z ∈ ball y r ⟷ z ∈ ball x r⟩ for z
    using ultrametric-dist-triangle[of x z y]
    * ultrametric-dist-triangle[of y z x]
    by (intro iffI) (simp-all add: dist-commute)
qed

```

```

lemma ultrametric-every-point-of-cball-is-centre :
  ⟨cball y r = cball x r⟩ if ⟨y ∈ cball x r⟩
proof (unfold set-eq-iff, rule allI)
  from ⟨y ∈ cball x r⟩ have * : ⟨dist x y ≤ r⟩ by simp
  show ⟨z ∈ cball y r ⟷ z ∈ cball x r⟩ for z
    using ultrametric-dist-triangle[of x z y]
    * ultrametric-dist-triangle[of y z x]
    by (intro iffI) (simp-all add: dist-commute)
qed

```

end

2.2 Balls are “clopen”

Balls are both open and closed.

context fixes $x :: \langle 'a :: \text{ultrametric-space} \rangle$ **begin**

lemma *ultrametric-open-cball* [intro, simp] : $\langle \text{open } (\text{cball } x \ r) \rangle$ **if** $\langle 0 < r \rangle$

proof (rule *openI*)

fix y **assume** $\langle y \in \text{cball } x \ r \rangle$

hence $\langle \text{cball } y \ r = \text{cball } x \ r \rangle$

by (rule *ultrametric-every-point-of-cball-is-centre*)

hence $\langle \text{ball } y \ (r / 2) \subseteq \text{cball } x \ r \rangle$

by (*metis ball-subset-cball cball-divide-subset-numeral subset-trans*)

moreover have $\langle 0 < r / 2 \rangle$ **by** (*simp add: 0 < r*)

ultimately show $\langle \exists e > 0. \text{ball } y \ e \subseteq \text{cball } x \ r \rangle$ **by** *blast*

qed

lemma $\langle \text{closed } (\text{cball } y \ r) \rangle$ **by** (*fact closed-cball*)

lemma *ultrametric-closed-ball* [intro, simp]: $\langle \text{closed } (\text{ball } x \ r) \rangle$ **if** $\langle 0 \leq r \rangle$

proof (*cases 0 < r*)

show $\langle r = 0 \implies \text{closed } (\text{ball } x \ r) \rangle$ **by** *simp*

next

assume $\langle r \neq 0 \rangle$

with $\langle 0 \leq r \rangle$ **have** $\langle 0 < r \rangle$ **by** *simp*

show $\langle \text{closed } (\text{ball } x \ r) \rangle$

proof (*unfold closed-def*)

have $\langle - \text{ball } x \ r = (\bigcup z \in - \text{ball } x \ r. \text{ball } z \ r) \rangle$

proof (*intro subset-antisym subsetI*)

from $\langle 0 < r \rangle$ **show** $\langle z \in - \text{ball } x \ r \implies z \in (\bigcup z \in - \text{ball } x \ r. \text{ball } z \ r) \rangle$ **for** z

by (*meson UN-iff centre-in-ball*)

next

show $\langle z \in (\bigcup z \in - \text{ball } x \ r. \text{ball } z \ r) \implies z \in - \text{ball } x \ r \rangle$ **for** z

by *simp (metis ComplD dist-commute mem-ball*

ultrametric-every-point-of-ball-is-centre)

qed

show $\langle \text{open } (- \text{ball } x \ r) \rangle$ **by** (*subst 0 < r, rule open-Union, simp*)

qed

qed

lemma *ultrametric-open-sphere* [intro, simp] : $\langle 0 < r \implies \text{open } (\text{sphere } x \ r) \rangle$

by (*fold cball-diff-eq-sphere*) (*simp add: open-Diff order-le-less*)

lemma *closed-sphere* [*intro, simp*] : $\langle \text{closed (sphere } y \ r) \rangle$
by (*metis open-ball cball-diff-eq-sphere closed-Diff closed-ccall*)

end

2.3 Balls are disjoint or contained

context *fixes* $x :: \langle 'a :: \text{ultrametric-space} \rangle$ **begin**

lemma *ultrametric-ball-ball-disjoint-or-subset*:
 $\langle \text{ball } x \ r \cap \text{ball } y \ s = \{\} \vee \text{ball } x \ r \subseteq \text{ball } y \ s \vee$
 $\text{ball } y \ s \subseteq \text{ball } x \ r \rangle$
proof (*unfold disj-imp, intro impI*)
assume $\langle \text{ball } x \ r \cap \text{ball } y \ s \neq \{\} \rangle \langle \neg \text{ball } x \ r \subseteq \text{ball } y \ s \rangle$
from $\langle \text{ball } x \ r \cap \text{ball } y \ s \neq \{\} \rangle$
obtain z **where** $\langle z \in \text{ball } x \ r \rangle$ **and** $\langle z \in \text{ball } y \ s \rangle$ **by** *blast*
with *ultrametric-every-point-of-ball-is-centre*
have $\langle \text{ball } x \ r = \text{ball } z \ r \rangle$ **and** $\langle \text{ball } y \ s = \text{ball } z \ s \rangle$ **by** *auto*
with $\langle \neg \text{ball } x \ r \subseteq \text{ball } y \ s \rangle$ **have** $\langle s < r \rangle$ **by** *auto*
with $\langle \text{ball } x \ r = \text{ball } z \ r \rangle$ **and** $\langle \text{ball } y \ s = \text{ball } z \ s \rangle$
show $\langle \text{ball } y \ s \subseteq \text{ball } x \ r \rangle$ **by** *auto*
qed

lemma *ultrametric-ball-ccall-disjoint-or-subset*:
 $\langle \text{ball } x \ r \cap \text{ccall } y \ s = \{\} \vee \text{ball } x \ r \subseteq \text{ccall } y \ s \vee$
 $\text{ccall } y \ s \subseteq \text{ball } x \ r \rangle$
proof (*unfold disj-imp, intro impI*)
assume $\langle \text{ball } x \ r \cap \text{ccall } y \ s \neq \{\} \rangle \langle \neg \text{ball } x \ r \subseteq \text{ccall } y \ s \rangle$
from $\langle \text{ball } x \ r \cap \text{ccall } y \ s \neq \{\} \rangle$
obtain z **where** $\langle z \in \text{ball } x \ r \rangle$ **and** $\langle z \in \text{ccall } y \ s \rangle$ **by** *blast*
with *ultrametric-every-point-of-ball-is-centre*
ultrametric-every-point-of-ccall-is-centre
have $\langle \text{ball } x \ r = \text{ball } z \ r \rangle \langle \text{ccall } y \ s = \text{ccall } z \ s \rangle$ **by** *blast+*
with $\langle \neg \text{ball } x \ r \subseteq \text{ccall } y \ s \rangle$ **have** $\langle s < r \rangle$ **by** *auto*
with $\langle \text{ball } x \ r = \text{ball } z \ r \rangle$ **and** $\langle \text{ccall } y \ s = \text{ccall } z \ s \rangle$
show $\langle \text{ccall } y \ s \subseteq \text{ball } x \ r \rangle$ **by** *auto*
qed

corollary *ultrametric-ccall-ball-disjoint-or-subset*:
 $\langle \text{ccall } x \ r \cap \text{ball } y \ s = \{\} \vee \text{ccall } x \ r \subseteq \text{ball } y \ s \vee$
 $\text{ball } y \ s \subseteq \text{ccall } x \ r \rangle$
using *Elementary-Ultrametric-Spaces.ultrametric-ball-ccall-disjoint-or-subset*
by *blast*

lemma *ultrametric-ccall-ccall-disjoint-or-subset*:
 $\langle \text{ccall } x \ r \cap \text{ccall } y \ s = \{\} \vee \text{ccall } x \ r \subseteq \text{ccall } y \ s \vee$
 $\text{ccall } y \ s \subseteq \text{ccall } x \ r \rangle$
proof (*unfold disj-imp, intro impI*)
assume $\langle \text{ccall } x \ r \cap \text{ccall } y \ s \neq \{\} \rangle \langle \neg \text{ccall } x \ r \subseteq \text{ccall } y \ s \rangle$

```

from  $\langle cball\ x\ r \cap cball\ y\ s \neq \{\} \rangle$ 
obtain  $z$  where  $\langle z \in cball\ x\ r \rangle$  and  $\langle z \in cball\ y\ s \rangle$  by blast
with ultrametric-every-point-of-cball-is-centre
have  $\langle cball\ x\ r = cball\ z\ r \rangle$   $\langle cball\ y\ s = cball\ z\ s \rangle$  by auto
with  $\langle \neg cball\ x\ r \subseteq cball\ y\ s \rangle$  have  $\langle s < r \rangle$  by auto
with  $\langle cball\ x\ r = cball\ z\ r \rangle$  and  $\langle cball\ y\ s = cball\ z\ s \rangle$ 
show  $\langle cball\ y\ s \subseteq cball\ x\ r \rangle$  by auto
qed

end

```

2.4 Distance to a Ball

```

context fixes  $a :: \langle 'a :: ultrametric-space \rangle$  begin

```

```

lemma ultrametric-equal-distance-to-ball:
 $\langle dist\ a\ y = dist\ a\ z \rangle$  if  $\langle a \notin ball\ x\ r \rangle$   $\langle y \in ball\ x\ r \rangle$   $\langle z \in ball\ x\ r \rangle$ 
proof (rule order-antisym)
show  $\langle dist\ a\ y \leq dist\ a\ z \rangle$ 
  by (rule order-trans[OF ultrametric-dist-triangle[of a y z]], simp)
  (metis dist-commute dual-order.strict-trans2 linorder-linear mem-ball that ultrametric-every-point-of-ball-is-centre)
next
show  $\langle dist\ a\ z \leq dist\ a\ y \rangle$ 
  by (rule order-trans[OF ultrametric-dist-triangle[of a z y]], simp)
  (metis dist-commute dual-order.strict-trans2 linorder-linear mem-ball that ultrametric-every-point-of-ball-is-centre)
qed

```

```

lemma ultrametric-equal-distance-to-cball:
 $\langle dist\ a\ y = dist\ a\ z \rangle$  if  $\langle a \notin cball\ x\ r \rangle$   $\langle y \in cball\ x\ r \rangle$   $\langle z \in cball\ x\ r \rangle$ 
proof (rule order-antisym)
show  $\langle dist\ a\ y \leq dist\ a\ z \rangle$ 
  by (rule order-trans[OF ultrametric-dist-triangle[of a y z]], simp)
  (metis dist-commute dual-order.trans linorder-linear mem-cball that ultrametric-every-point-of-cball-is-centre)
next
show  $\langle dist\ a\ z \leq dist\ a\ y \rangle$ 
  by (rule order-trans[OF ultrametric-dist-triangle[of a z y]], simp)
  (metis dist-commute dual-order.trans linorder-linear mem-cball that ultrametric-every-point-of-cball-is-centre)
qed

```

```

end

```

```

context fixes  $x :: \langle 'a :: ultrametric-space \rangle$  begin

```

lemma *ultrametric-equal-distance-between-ball-ball*:
 $\langle ball\ x\ r \cap ball\ y\ s = \{\} \implies$
 $\exists d. \forall a \in ball\ x\ r. \forall b \in ball\ y\ s. dist\ a\ b = d \rangle$
by (*metis disjoint-iff dist-commute ultrametric-equal-distance-to-ball*)

lemma *ultrametric-equal-distance-between-ball-cball*:
 $\langle ball\ x\ r \cap cball\ y\ s = \{\} \implies$
 $\exists d. \forall a \in ball\ x\ r. \forall b \in cball\ y\ s. dist\ a\ b = d \rangle$
by (*metis disjoint-iff dist-commute ultrametric-equal-distance-to-ball*
ultrametric-equal-distance-to-cball)

lemma *ultrametric-equal-distance-between-cball-ball*:
 $\langle cball\ x\ r \cap ball\ y\ s = \{\} \implies$
 $\exists d. \forall a \in cball\ x\ r. \forall b \in ball\ y\ s. dist\ a\ b = d \rangle$
by (*metis disjoint-iff-not-equal dist-commute ultrametric-equal-distance-to-ball*
ultrametric-equal-distance-to-cball)

lemma *ultrametric-equal-distance-between-cball-cball*:
 $\langle cball\ x\ r \cap cball\ y\ s = \{\} \implies$
 $\exists d. \forall a \in cball\ x\ r. \forall b \in cball\ y\ s. dist\ a\ b = d \rangle$
by (*metis disjoint-iff dist-commute ultrametric-equal-distance-to-cball*)

end

3 Additional Properties

Here are a few other interesting properties.

3.1 Cauchy Sequences

lemma (*in ultrametric-space*) *ultrametric-dist-triangle-generalized*:
 $\langle n < m \implies dist\ (\sigma\ n)\ (\sigma\ m) \leq (MAX\ l \in \{n..m - 1\}. dist\ (\sigma\ l)$
 $(\sigma\ (Suc\ l))) \rangle$
proof (*induct m*)
show $\langle n < 0 \implies dist\ (\sigma\ n)\ (\sigma\ 0) \leq (MAX\ l \in \{n..0 - 1\}. dist\ (\sigma\ l)$
 $(\sigma\ (Suc\ l))) \rangle$ **by** *simp*
next
case (*Suc m*)
show $\langle dist\ (\sigma\ n)\ (\sigma\ (Suc\ m)) \leq (MAX\ l \in \{n..Suc\ m - 1\}. dist\ (\sigma\ l)$
 $(\sigma\ (Suc\ l))) \rangle$
proof (*cases n = m*)
show $\langle n = m \implies dist\ (\sigma\ n)\ (\sigma\ (Suc\ m)) \leq (MAX\ l \in \{n..Suc\ m$
 $- 1\}. dist\ (\sigma\ l)\ (\sigma\ (Suc\ l))) \rangle$
by *simp*
next
assume $\langle n \neq m \rangle$
with $\langle n < Suc\ m \rangle$ **obtain** m' **where** $\langle m = Suc\ m' \rangle \langle n \leq m' \rangle$

by (*metis le-add1 less-Suc-eq less-imp-Suc-add*)
have $\langle \{n..Suc\ m'\} = \{n..m-1\} \cup \{m\} \rangle$
by (*simp add: $\langle m = Suc\ m' \rangle \langle n \leq m' \rangle$ atLeastAtMostSuc-conv le-Suc-eq*)
have $\langle dist\ (\sigma\ n)\ (\sigma\ (Suc\ m)) \leq \max\ (dist\ (\sigma\ n)\ (\sigma\ m))\ (dist\ (\sigma\ m)\ (\sigma\ (Suc\ m))) \rangle$
by (*simp add: ultrametric-dist-triangle*)
also have $\langle \dots \leq \max\ ((MAX\ l \in \{n..m-1\}. dist\ (\sigma\ l)\ (\sigma\ (Suc\ l)))\ (dist\ (\sigma\ m)\ (\sigma\ (Suc\ m))) \rangle$
using *Suc.hyps Suc.premis $\langle n \neq m \rangle$ by linarith*
also have $\langle \dots = (MAX\ l \in \{n..Suc\ m-1\}. dist\ (\sigma\ l)\ (\sigma\ (Suc\ l))) \rangle$
by (*subst Max-Un[of - $\langle (\lambda l. dist\ (\sigma\ l)\ (\sigma\ (Suc\ l)) \rangle \text{ ' } \{m\} \rangle$, simplified, symmetric]*)
(simp-all add: $\langle m = Suc\ m' \rangle \langle n \leq m' \rangle \langle \{n..Suc\ m'\} = \{n..m-1\} \cup \{m\} \rangle$)
finally show $\langle dist\ (\sigma\ n)\ (\sigma\ (Suc\ m)) \leq (MAX\ l \in \{n..Suc\ m-1\}. dist\ (\sigma\ l)\ (\sigma\ (Suc\ l))) \rangle$.
qed
qed

lemma (*in ultrametric-space*) *ultrametric-Cauchy-iff:*
 $\langle Cauchy\ \sigma \longleftrightarrow (\lambda n. dist\ (\sigma\ (Suc\ n))\ (\sigma\ n)) \longrightarrow 0 \rangle$
proof (*rule iffI*)
assume $\langle Cauchy\ \sigma \rangle$
show $\langle (\lambda n. dist\ (\sigma\ (Suc\ n))\ (\sigma\ n)) \longrightarrow 0 \rangle$
proof (*unfold lim-sequentially, intro allI impI*)
fix $\varepsilon :: real$
assume $\langle 0 < \varepsilon \rangle$
from $\langle Cauchy\ \sigma \rangle$ [*unfolded Cauchy-altdef, rule-format, OF $\langle 0 < \varepsilon \rangle$*]
show $\langle \exists N. \forall n \geq N. dist\ (dist\ (\sigma\ (Suc\ n))\ (\sigma\ n))\ 0 < \varepsilon \rangle$
by (*auto simp add: dist-commute*)
qed
next
assume *convergent* : $\langle (\lambda n. dist\ (\sigma\ (Suc\ n))\ (\sigma\ n)) \longrightarrow 0 \rangle$
show $\langle Cauchy\ \sigma \rangle$
proof (*unfold Cauchy-altdef2, intro allI impI*)
fix $\varepsilon :: real$
assume $\langle 0 < \varepsilon \rangle$
from *convergent* [*unfolded lim-sequentially, rule-format, OF $\langle 0 < \varepsilon \rangle$*]
obtain N **where** $*$: $\langle N \leq n \implies dist\ (\sigma\ (Suc\ n))\ (\sigma\ n) < \varepsilon \rangle$ **for** n
by (*simp add: dist-real-def*) *blast*
have $\langle N < n \implies dist\ (\sigma\ n)\ (\sigma\ N) < \varepsilon \rangle$ **for** n
proof (*subst dist-commute, rule le-less-trans*)
show $\langle N < n \implies dist\ (\sigma\ N)\ (\sigma\ n) \leq (MAX\ l \in \{N..n-1\}. dist$

```

(σ l) (σ (Suc l)))
  by (fact ultrametric-dist-triangle-generalized)
next
  show ⟨N < n ⟹ (MAX l∈{N..n - 1}. dist (σ l) (σ (Suc l)))
< ε⟩
  by simp (metis * atLeastAtMost-iff dist-commute)
qed
with ⟨0 < ε⟩ have ⟨N ≤ n ⟹ dist (σ n) (σ N) < ε⟩ for n
  by (cases ⟨N = n⟩) simp-all
thus ⟨∃ N. ∀ n ≥ N. dist (σ n) (σ N) < ε⟩ by blast
qed
qed

```

3.2 Isosceles Triangle Principle

```

lemma (in ultrametric-space) ultrametric-isosceles-triangle-principle
:
  ⟨dist x z = max (dist x y) (dist y z)⟩ if ⟨dist x y ≠ dist y z⟩
proof (rule order-antisym)
  show ⟨dist x z ≤ max (dist x y) (dist y z)⟩
    by (fact ultrametric-dist-triangle)
next
  from ⟨dist x y ≠ dist y z⟩ linorder-less-linear
  have ⟨dist x y < dist y z ∨ dist y z < dist x y⟩ by blast
  with ultrametric-dist-triangle[of y z x]
    ultrametric-dist-triangle[of x y z]
  show ⟨max (dist x y) (dist y z) ≤ dist x z⟩
    by (elim disjE) (simp-all add: dist-commute)
qed

```

3.3 Distance to a convergent Sequence

```

lemma ultrametric-dist-to-convergent-sequence-is-eventually-const :
  fixes σ :: ⟨nat ⟹ 'a :: ultrametric-space⟩
  assumes ⟨σ ⟶ Σ⟩ and ⟨x ≠ Σ⟩
  shows ⟨∃ N. ∀ n ≥ N. dist (σ n) x = dist Σ x⟩
proof -
  from ⟨x ≠ Σ⟩ have ⟨0 < dist x Σ⟩ by simp
  then obtain ε where ⟨0 < ε⟩ ⟨ball x ε ∩ ball Σ ε = {}⟩
    by (metis centre-in-ball disjoint-iff mem-ball order-less-le
ultrametric-every-point-of-ball-is-centre)

  from ⟨σ ⟶ Σ⟩ ⟨0 < ε⟩ obtain N where ⟨N ≤ n ⟹ σ n ∈
ball Σ ε⟩ for n
    by (auto simp add: dist-commute lim-sequentially)
  with ⟨0 < ε⟩ ⟨ball x ε ∩ ball Σ ε = {}⟩ have ⟨N ≤ n ⟹ dist (σ
n) x = dist Σ x⟩ for n
    by (metis centre-in-ball dist-commute ultrametric-equal-distance-between-ball-ball)
  thus ⟨∃ N. ∀ n ≥ N. dist (σ n) x = dist Σ x⟩ by blast
qed

```

3.4 Diameter

lemma *ultrametric-diameter* : $\langle \text{diameter } S = (\text{SUP } y \in S. \text{dist } x \ y) \rangle$
 if $\langle \text{bounded } S \rangle$ and $\langle x \in S \rangle$ for $x :: \langle 'a :: \text{ultrametric-space} \rangle$
 proof –
 from $\langle x \in S \rangle$ have $\langle S \neq \{\} \rangle$ by *blast*
 show $\langle \text{diameter } S = (\text{SUP } y \in S. \text{dist } x \ y) \rangle$
 proof (rule *order-antisym*)
 from *diameter-bounded-bound*[*OF* $\langle \text{bounded } S \rangle \langle x \in S \rangle$]
 have $\langle y \in S \implies \text{dist } x \ y \leq \text{diameter } S \rangle$ for y .
 thus $\langle (\text{SUP } y \in S. \text{dist } x \ y) \leq \text{diameter } S \rangle$
 by (rule *cSUP-least*[*OF* $\langle S \neq \{\} \rangle$])
 next
 have $\langle \text{bdd-above } (\text{dist } x \ ' S) \rangle$
 by (*meson bdd-above.I2 bounded-any-center* $\langle \text{bounded } S \rangle$)
 have $\langle y \in S \implies z \in S \implies \text{dist } y \ z \leq \max (\text{dist } x \ y) (\text{dist } x \ z) \rangle$
 for $y \ z$
 by (*metis dist-commute ultrametric-dist-triangle*)
 also have $\langle y \in S \implies z \in S \implies$
 $\max (\text{dist } x \ y) (\text{dist } x \ z) \leq (\text{SUP } y \in S. \text{dist } x \ y) \rangle$ for $y \ z$
 by (*cases* $\langle \text{dist } x \ y \leq \text{dist } x \ z \rangle$)
 (*simp-all add: cSUP-upper2*[*OF* $\langle \text{bdd-above } (\text{dist } x \ ' S) \rangle$])
 finally have $* : \langle y \in S \implies z \in S \implies \text{dist } y \ z \leq (\text{SUP } y \in S.$
 $\text{dist } x \ y) \rangle$ for $y \ z$.
 have $\langle (\text{SUP } (y, z) \in S \times S. \text{dist } y \ z) \leq (\text{SUP } y \in S. \text{dist } x \ y) \rangle$
 by (rule *cSUP-least*) (use $*$ in $\langle \text{auto simp add: } \langle S \neq \{\} \rangle$)
 thus $\langle \text{diameter } S \leq \text{Sup } (\text{dist } x \ ' S) \rangle$
 by (*simp add: diameter-def* $\langle S \neq \{\} \rangle$)
 qed
 qed

3.5 Totally disconnected

lemma *ultrametric-totally-disconnected* :
 $\langle \exists x. S = \{x\} \rangle$ if $\langle S \neq \{\} \rangle \langle \text{connected } S \rangle$
 for $S :: \langle 'a :: \text{ultrametric-space set} \rangle$
 proof –
 from $\langle S \neq \{\} \rangle$ obtain x where $\langle x \in S \rangle$ by *blast*
 have $\langle \text{ball } x \ r \cap S = \{\} \vee - \text{ball } x \ r \cap S = \{\} \rangle$ if $\langle 0 < r \rangle$ for r
 by (rule $\langle \text{connected } S \rangle$ [*unfolded connected-def, simplified, rule-format*])
 (*simp-all, use order-less-imp-le that ultrametric-closed-ball in blast*)
 with $\langle x \in S \rangle$ have $\langle 0 < r \implies - \text{ball } x \ r \cap S = \{\} \rangle$ for r
 by (*metis centre-in-ball disjoint-iff*)
 hence $\langle 0 < r \implies y \in S \implies \text{dist } x \ y < r \rangle$ for $r \ y$
 by (*auto simp add: disjoint-iff*)
 hence $\langle y \in S \implies \text{dist } x \ y = 0 \rangle$ for y
 by (*metis dist-self order-less-irrefl zero-less-dist-iff*)
 hence $\langle y \in S \implies y = x \rangle$ for y by *simp*
 with $\langle x \in S \rangle$ show $\langle \exists x. S = \{x\} \rangle$ by *blast*

qed